

ŁAMIGŁÓWKI I PROGRAMY KOMPUTEROWE

Nr 1

Cena 2200 zł

S P I S T R E S C I

OD REDAKCJI	2
ŁAMIGŁÓWKI MATEMATYCZNE, ZGADYWANKI	2 - 5
Zadanie dla komputera	2
Liczby Fibonacciego	3
Komputrus - rzymski matematyk	3 - 4
Teoria względności Einsteina i masa	4
Egzotyczna flaga	5
Popularna melodia jednogłosowa	5
ZADANIE TEKSTOWE dla komputera	6
WARI - gra planszowa dla 2 osób	6 - 8
CZOŁG NA MOŚCIE - gra zręcznościowa >na refleks<	9 - 11
FAJKA RYBAKA - edukacja szkolna - liczby względne	11 - 16
ROZWIĄZANIA ZADAN	15
W NASTĘPNYM NUMERZE	15
CIEKAWA GRAFIKA	16

GRA AFRYKAŃSKA >W A R I<

 GR-A: Jacek
 GR-B: Agata

 PUNKTY GR-A
 PUNKTY GR-B

GRA NR.1

PREMIA 3 PKT.

 GRA
 PLAN-
 SZOWA
 STR 6

	12	11	10	9	8	7	<B
7	0	0	3	7	8	1	
	8	8	0	0	0	1	
A>	1	2	3	4	5	6	2

SZANOWNI CZYTELNICY

Jest w Polsce ponad pół miliona mikrokomputerów ZX SPECTRUM i kompatybilnych TIMEX, JUNIOR oraz UNIPOLBRIT.

Sprawdzone zostały z zagranicy, względnie wyprodukowane w kraju (JUNIOR, POLBRIT) w latach 1985-1989.

W większości stanowią własność prywatną, w znacznej mierze znajdują się różnie na wyposażeniu szkół podstawowych i średnich.

W szczególności, z wprowadzeniem mikrokomputerów do szkół związane były duże nadzieje na ich szerokie wykorzystanie do dydaktyki szkolnej.

Z perspektywy lat ocenić jednakże można, że wykorzystanie mikrokomputerów SPECTRUM lub innych w zakresie dydaktyki dalekie jest od zapotrzebowania i potencjalnych możliwości.

Zbyt mało bowiem jest w obiegu programów dydaktycznych. Nie brak jedynie angielskojęzycznych programów zręcznościowych, w których się strzela, bombarduje, walczy z groźnymi stworami itp. Nasze nowe czasopismo ŁAMIGŁÓWKI I PROGRAMY KOMPUTEROWE stawia sobie za cel częściowe wypełnienie tej luki dydaktycznej.

Zamierza też służyć wszystkim prywatnym użytkownikom, a również szkołom i klubom komputerowym w bardziej owocnym zagospodarowaniu posiadanych mikrokomputerów SPECTRUM i pokrewnych typów.

Pismo prezentować będzie programy i programiki z zakresu matematyki, fizyki, geografii, edukacji ekonomicznej, również gry planszowe, gry logiczne, graficzne, muzyczne itd.

Uczyć będziemy opracowywania programów na własny użytek - do rozwiązywania zadań matematycznych, prowadzenia budżetu domowego, oceny uczniów w klasie, ewidencji imion i nazwisk itp.

Programy pisane będą w języku BASIC - SPECTRUM, jako najpopularniejszym i najłatwiejszym do zastosowania w ramach amatorskich.

W miarę unikać będziemy instrukcji właściwych tylko dla SPECTRUM, aby z pisma mogli korzystać również posiadacze ATARI i COMMODORE.

Część materiału będzie podawana w formie konkretnych zadań i zgadywanek, przy czym rozwiązania zamieszczane będą na przedostatniej stronie czasopisma.

Życzymy czytelnikom miłej i użytecznej rozrywki.

ŁAMIGŁÓWKI MATEMATYCZNE - ZGADYWANKI

ZADANIE DLA KOMPUTERA

```

RAZ          Program 1
RAZ          10 FOR R=8 TO 9
RAZ          20 FOR A=0 TO 9
RAZ          30 FOR Z=1 TO 9
RAZ          40 LET TUZIN=1200*R+120*A+12*Z
RAZ          50 GOSUB 100
RAZ          60 IF X=Z THEN GO TO 80
RAZ          70 NEXT Z: NEXT A:NEXT R
RAZ          80 PRINT AT 10,4;R;A;Z;"*12=";TUZIN
RAZ          90 STOP
RAZ          100 LET WS= STR$ TUZIN
+ RAZ        110 LET XS=WS(3)
-----
RAZ          120 LET X= VAL XS
TUZIN        130 RETURN

```

W zadaniu po lewej stronie (z >SAM na SAM< nr.5/91) każdej literze odpowiada określona cyfra. Należy znaleźć te cyfry.

Sposób jest jeden - trzeba pod poszczególne litery podstawiać co raz to inne liczby od 0 do 9. Jest to syzyfowa praca i aż prosi się aby skorzystać z komputera, który bardzo takie rachunki lubi. Przeprowadzi je w niespełną minutę, jeśli wpiszemy do niego powyższy programik.

INTERPRETACJA PROGRAMU: W liniach 10-70 zamieszczone są 3 pętle, podstawiające wartości liczbowe w miejsce liter R, A i Z.

W linii 40 jest przedstawiona liczba TUZIN. W linii 60 zaś zawarty jest warunek dany w tekście zadania, że trzecia cyfra sumy (oznaczona w programie przez X) równa jest cyfrze Z mnożnika. W linii 80 wydruk rozwiązania, a w podprogramie 100-120 manipulacje z określeniem 3-ciej cyfry liczby TUZIN. Napisanie programiku zajęło opracowującemu około 10 minut, komputer liczył minutę - chyba jednak szybciej niż >na piechotę<.

LICZBY FIBONACCIEGO

Liczby Fibonacciego to ciąg liczb naturalnych, z których każda (z wyjątkiem pierwszych dwóch) równa jest sumie dwu poprzednich. Zakłada się przy tym, że pierwsza i druga liczba równe są jedności. A więc trzecia wynosi 2, czwarta 3, następne kolejno 5, 8, 13, 21, 34, itd.

Spróbuj obliczyć wartość 39-mej liczby ciągu Fibonacciego. Rozwiązanie zadania >na piechotę< na kartce papieru może być nawet zbyt długo nie trwało, ale jakże łatwo o pomyłkę, gdy operuje się milionowymi liczbami.

A komputer się nie pomyli i poda rozwiązanie po kilku sekundach po wpisaniu programu nr 2. Przy okazji określi nie tylko szukaną liczbę, ale wszystkie liczby Fibonacciego w zadanym przedziale.

Program 2

```
10 PRINT PAPER 6; INK 9;"          LICZBY FIBONACCIEGO      "  
20 INPUT "Podaj kolejne liczby (od-do):","m=";m,"n=";n  
30 PRINT "Liczby F. od ";m;"-tej do ";n;"-tej"  
40 DIM F(n+2): LET F(1)=1: LET F(2)=1  
50 FOR k=1 TO n: LET F(k+2)=F(k+1)+F(k)  
60 IF k=m THEN PRINT k, INK 2:F(k)  
70 NEXT k: PRINT "-----": GO TO 20
```

KOMPUTRUS

Miesięcznik >KOMPUTER< w nr 18 z 1988 r. zamieścił bardzo sympatyczne matematyczne zadanie konkursowe:

>Gdy Cesarz Justynian zakazał uprawiania matematyki - matematycy rzymscy, którym groziła niewola, postanowili popełnić zbiorowe samobójstwo.

Mieli stanąć w koło, poczem odliczać ustaloną liczbę i zabijać tego na kogo wypadnie.

Jeden z matematyków, Komputrus, wolał niewolę od śmierci. Postanowił więc stanąć w szeregu (kole), by być ostatnim.

Oblicz miejsce Komputrusa, przyjmując, że wszystkich matematyków było 32, a liczbą odliczania 7.<

Zadanie rozwiązać można na kartce papieru, rysując np. 32 kreski i następnie, odliczając je >w kołko<, skreślać co 7-me.

Dla cierpliwych nie potrwa to zbyt długo, ale jak łatwo o pomyłkę. A gdyby zdesperowanych matematyków było więcej i krok odliczania też znacznie większy?

Lepiej więc odwołać się do pomocy komputera.

Przykładowy programik 3 oblicza ostatnie miejsce dla Komputrusa przy dowolnej liczbie matematyków i dowolnym kroku odliczania.

Konstrukcja programu jest bardzo prosta. Komputer faktycznie odlicza liczby jedna po drugiej w pętli (linijki 70-100) wg zada-

nego kroku i zeruje liczbę, na które >wypadnie<, tak, że nie biorą one udziału w dalszym odliczaniu (linia 90). Ostatecznie na placu boju pozostaje jedyna ostatnia liczba (linia 80), czyli miejsce, na którym powinien stanąć Komputrus.

Program 3

```
10 CLS : PRINT PAPER 6; INK 0;"           MIEJSCE KOMPUTRUSA
20 INPUT "Podaj liczbę matematyków m=";m
30 INPUT "Podaj krok odliczania k=";k
40 DIM a(m): LET w=1: LET s=0
50 FOR l=1 TO m: LET a(l)=l: NEXT l
60 PRINT : PRINT "Dla m=";m;" matematyków i kroku""odliczania
k=";k;". Komputrus"" powinien stanąć na miejscu ";
70 FOR i=1 TO m: IF a(i)=0 THEN GO TO 100
80 LET s=s+1: IF w=m THEN PRINT INK 2;a(i): PRINT : GO TO 20
90 IF s=k THEN LET a(i)=0: LET s=0: LET w=w+1
100 NEXT i: GO TO 70
```

TEORIA WZGLĘDNOŚCI I MASA

Według teorii względności Einsteina masa ciała będącego w ruchu zmienia się wraz z prędkością.

Przy jakiej prędkości masa wzrasta 12-krotnie w stosunku do masy ciała będącego w spoczynku (z dokładnością do 100 km/sek) ?.

Poniższy program nr. 4 rozwiązuje zadanie w ciągu kilkunastu sekund. Wykonuje dodatkowo wykres przyrostu masy i podaje tabelkę wzrostu masy 2,3,4...12 krotnie, wraz z odpowiednimi prędkościami. Linie 10 - 140 rysują grafikę planszy, zaś wykres wzrostu masy ciała, związanego ze wzrostem prędkości, realizowany jest w liniach 150 - 190.

Podprogram 200 - 210 drukuje prędkości, przy których masa ciała wynosi 2,3,4...12 krotności masy spoczynkowej.

Program 4

```
10 PRINT PAPER2; INK 9;"   TEORIA WZGLĘDNOŚCI I MASA   "
20 PRINT : PRINT " WZROST   PREDKOŚCI"" MASY   TYS KM/SEK"
30 PRINT INK 2;"-----"
40 BORDER 4: LET b=4: LET m=1
50 FOR i=1 TO 2: PRINT PAPER 6; AT 19+i,0;"
   ": NEXT i
60 FOR i=0 TO 25
70 IF INT (i/5)=i/5 THEN PLOT 10*i,15: DRAW 0,-14: GO TO 90
80 PLOT 10*i,15: DRAW 0,-5
90 NEXT i
100 FOR i=0 TO 4: PRINT PAPER 6; AT 21,3+i*6;260+10*i: NEXT i
110 FOR i=1 TO 15
120 IF INT (i/5)=i/5 THEN PLOT 250,17+5*i: DRAW 5,0: GO TO 140
130 PLOT 250,17+5*i: DRAW 2,0
140 NEXT i
150 FOR v=2500 TO 2990
160 LET w=v*v/9e6: LET u=1/SQR (1-w)
170 PLOT INK 2;v/2-1250,5*u+16
180 IF INT u=m+1 THEN GO SUB 200: DRAW INK 2;0,-5*u
190 NEXT v: STOP
```



```

200 LET b=b+1: LET m=m+1
210 PRINT AT b,4-LEN STR$ m;m;TAB 12;v/10
220 RETURN

```

EGZOTYCZNA FLAGA

Poniższy program nr. 5 rysuje kolorową flagę niedużego państwa afrykańskiego. Jakie to państwo ?

```

Program 5
10 FOR j=1 TO 5
20 FOR i=2*j TO 2*j+1
30 IF INT (j/2)=j/2 THEN GO TO 50
40 PRINT PAPER 4;AT i,8;"          ": GO TO 60
50 PRINT PAPER 6;AT i,8;"          "
60 NEXT i: NEXT j
70 FOR i=2 TO 7: PRINT AT i,8; PAPER 2;"          ": NEXT i
80 INK 7: FOR i=-10 TO 10: PLOT 93,130: DRAW i,7: NEXT i
90 FOR i=4 TO 19: PLOT 85,126: DRAW 7,i: NEXT i
100 FOR i=4 TO 19: PLOT 100,126: DRAW -7,i: NEXT i
110 PRINT INK 2;AT 13,10;"FLAGA ";CHRS 84;CHRS 79;CHRS 71;CHRS
79: STOP

```

MELODIA JEDNOGŁOSOWA

Mikrokomputer SPECTRUM ma znacznie mniejsze możliwości akustyczne niż ATARI czy COMMODORE. Nie ma to jednakże większego znaczenia przy melodiach jednogłosowych.

Poniższy program nr. 6 przedstawia bardzo popularną melodię regionalnego tańca ludowego. Co to za melodia ?

Tonacja dźwięków i długość ich trwania określone zostały w DATA w liniijkach 50 - 100.

Jak łatwo zauważyć - przez odpowiednią zmianę cyfr w tych liniijkach można w program wpisać dowolne inne melodie. Trzeba przy tym zmienić w linii 15 wartość i, która określa ilość nutek melodii.

```

Program 6
10 RESTORE 50: FOR i=1 TO 88: READ m,n
20 IF m=50 THEN PAUSE 12.5: GO TO 40
25 IF m=100 THEN STOP
30 BEEP m/1.2,n
40 NEXT i
50 DATA .25,7,.25,10,.37,3,.12,7,.25,7,.5,5,50,0,.25,5,.25,8,.3
7,2,.12,5,.25,3,.5,7,50,0
60 DATA .25,7,.25,10,.37,3,.12,7,.25,7,.5,5,50,0,.25,12,.25,10,
.37,8,.12,2,.25,5,.5,3,50,0
70 DATA .25,3,.25,3,.37,15,.12,14,.25,14,.5,12,50,0,.25,12,.25,
15,.37,14,.12,12,.25,12,.5,10,50,0
80 DATA .25,7,.25,10,.37,3,.12,7,.25,7,.5,5,50,0,.25,12,.25,10,
.37,8,.12,2,.25,5,.5,3,50,0
90 DATA .25,3,.25,3,.37,15,.12,14,.25,14,.5,12,50,0,.25,12,.25,
15,.37,14,.12,12,.25,12,.5,10,50,0
100 DATA .25,7,.25,10,.37,3,.12,7,.25,7,.5,5,50,0,.25,12,.25,10,
.37,8,.12,2,.25,5,.5,15,100,0

```

Znaleźć pięć najmniejszych liczb naturalnych N takich, że między N oraz $N+10$ nie ma żadnej liczby pierwszej.

Zadanie zaczerpnięte zostało z byłego miesięcznika komputerowego >IKS<, który w ramach konkursu >Liga Myślących< zamieścił wiele oryginalnych zadań matematycznych (na komputer).

Rozwiązuje je poniższy programik, który sprawdza pracowicie warunki zadania pokolei dla wszystkich liczb poczynając od 110.

Program 7

```
10 LET n=110: LET l=0
20 LET n=n+1: LET m=n+10: IF l=5 THEN STOP
30 FOR i=n TO m: LET q=SQR i
40 FOR j=2 TO q: IF i/j=INT (i/j) THEN GO TO 60
50 NEXT j: GO TO 20
60 IF i=m THEN LET l=l+1: PRINT "Liczba szukana "; INK 2;"N";(
1); INK 0;"="; PAPER 2; INK 9;n: GO TO 20
70 NEXT i
```

GRA PLANSZOWA

W dziale tym, w kolejnych numerach >LAMIGŁÓWKI<, zamieszczać będziemy gry planszowe, do rozgrywania we dwie osoby, względnie do gry z komputerem.

Będą to częściowo gry bardzo popularne, wszystkie zamieszczone w miesięczniku rozrywkowym >SAM na SAM<, po ich przeróbce na programy komputerowe.

W bieżącym numerze prześliczna, wymagająca spostrzegawczości, afrykańska gra >WARI<, dla dwóch osób.

Zasady gry zawarte są w instrukcji w programie.

Na stronie tytułowej zamieszczono wydruk planszy gry, która jest oczywiście kolorowa, z migającymi kamykami, z komunikatami, punktacją itd.

Program 8

```
10>GO SUB 850:BORDER 0:PAPER 0:INK 7:CLS
12 REM * Instrukcja *
15 PRINT PAPER 6; INK 0;" GRA AFRYKAŃSKA >W A R I< "
20 PRINT : PRINT " JEST TO STARA GRA AFRYKAŃSKA. UDZIAŁ BIORĄ
2 OSOBY, ZAGRYWAJĄCNA PRZEMIANY."
22 PRINT " KAŻDY GRACZ MA DO DYSPOZYCJI 6PÓŁ PO 4 KAMIEŃ W
KAŻDYM. ZA-GRYWAJĄCY BIERZE WSZYSTKIE KA-MIENIE Z DOWOLNEGO
SWĘGO POLA I ROZSIEWA JE PO JEDNYM W KIE-RUNKU PRZEZ";
24 PRINT "CIWNYM DO RUCHU WSKA-ZÓWEK ZĘGARA, POCZYNAJĄC OD N
A-STEPNEGO POLA - RÓWNIET PO PO-LACH PRZECIWNIA."
26 PRINT " JESLI PO ROZSIANIU - DOŁOTENIUOSTATNIEGO KAMYKA W
POLU PRZECIWNIA ZNAJDUJĄ SIĘ ŁACZNIE 2 LUB 3 KAMIEŃ, WTEDY GR
ACZ ZAGRYWA-"
28 PRINT "JĄCY ZABIERA JE JAKO SVOJĄ ZDO-BYCZ I ODKŁADA NA PO
LU ZAPASOWYMP O SWEJ PRAWĘJ STRONIE."
30 PRINT #1; BRIGHT 1; FLASH 1; AT 1,3;"Nacisnij dowolny klawis
z": PAUSE 0: CLS
34 PRINT : PRINT " STRATY PRZECIWNIA MOGA BYĆ WIEKSZE O K
AMIEŃ PRZEDOSTAT-NIEGO I POPRZEDNICH PÓŁ, JESLI RÓWNIET I W
TYCH POLACH BĘDĄ ŁA-CZNIE PO 2 LUB 3 KAMYKI.
```

```

36 PRINT " KTO PIERWSZY ZBIERZE 24 KAMIE-NI
TEN GRE WYGRYWA. MOJNA GRAC DOWOLNA ILOSC GIER, WOWCZAS KOM-PUT
ER POKAZUJE NARASTAJACA ILOSC PUNKTOW DLA OBU GRACZY."
38 PRINT " JESLI GRACZ PODEJMUJE Z POLA WIECEJ NI 11 KAMIEN
I, TO POLE ZKTOREGO KAMIENIE ZOSTALY PODJETENIE JEST OBJETE ROZS
IEWANIEM."
40 PRINT #1: BRIGHT 1: FLASH 1;AT 1,3;"Nacisnij dowolny klawis
z": PAUSE 0: CLS
49 REM * Grafika planszy *
50 CLS : PRINT PAPER 6; INK 0;" GRA AFRYKANSKA >W A R I<
"
60 DIM a(13): DIM m(16): DIM n(29)
80 INPUT "PODAJ IMIONA GRAJACYCH","GR-A ";m$,"GR-B ";n$
85 PRINT AT 3,1;"GR-A:"; INK 6;m$
90 PRINT AT 4,1;"GR-B:"; INK 6;n$
95 INK 5: PLOT 6,164: DRAW 244,0
100 PRINT INK 5;AT 6,1;"GRA NR."
105 PLOT 6,116: DRAW 80,0: DRAW 0,16: DRAW -80,0: DRAW 0,-16
110 PRINT INK 7;AT 3,16;"PUNKTY GR-A:";AT 4,16;"PUNKTY GR-B"
130 PLOT 218,132: DRAW 32,0: DRAW 0,22: DRAW -32,0: DRAW 0,-22
140 PRINT PAPER 1; INK 7: BRIGHT 1;AT 9,4;" 12 11 10 9 8
7 ";AT 16,4;" 1 2 3 4 5 6 "
150 FOR l=0 TO 2: PLOT 32,48+l*24: DRAW 192,0: NEXT l
160 FOR l=0 TO 6: PLOT 32+32*l,48: DRAW 0,48: NEXT l
170 PLOT 2,72: DRAW 28,0: DRAW 0,32: DRAW -28,0: DRAW 0,-32
175 PLOT 254,72: DRAW 0,-32: DRAW -28,0: DRAW 0,32: DRAW 28,0
178 PRINT INK 7: BRIGHT 1;AT 16,2;"A:";AT 9,28;"<B"
180 PLOT 226,92: DRAW 18,0: DRAW 0,12: DRAW -214,0
185 PLOT 30,52: DRAW -18,0: DRAW 0,-12: DRAW 214,0
190 INK 7
200 LET gra=0: LET spktA=0: LET spktB=0
209 REM * Kolejna gra *
210 LET gra=gra+1: PRINT AT 6,9-LEN STR$ gra:gra
215 LET pktA=0: LET pktB=0: REM * Punkty (zdobyte kamienie) w k
olejnej grze *
220 GO SUB 400
240 INPUT "KTO ZACZYNA GRE?","A=";(m$),"B=";(n$);TAB 27;z$: BEE
P .1,16: BEEP .1,11
242 IF z$="A" OR z$="a" THEN LET r=1: GO TO 260
244 IF z$="B" OR z$="b" THEN LET r=2: GO TO 260
246 GO TO 240
250 LET r=r+1
255 REM * Kolejne ruchy *
260 IF INT (r/2)=r/2 THEN LET w=14: LET c$=n$: GO TO 265
262 LET w=13: LET c$=m$
265 INPUT "PODAJ, ";(c$);TAB 14;".NUMER POLA Z KTO-REGO ZABIERA
SZ KAMIENIE:";TAB 27;b$
266 IF CODE b$<49 OR CODE b$>57 THEN BEEP .1,-10: GO TO 265
267 LET b=VAL b$: IF b>12 THEN BEEP .1,-10: GO TO 265
268 BEEP .1,17: BEEP .1,14: LET c=b
270 IF w=13 AND b>6 THEN PRINT FLASH 1;AT 18,1;"UWATAJ. TWOJE
POLA MAJA NR 1-6": BEEP .2,-10: PAUSE 100: PRINT AT 18,0;"
": GO TO 265
275 IF w=14 AND b<7 THEN PRINT FLASH 1;AT 18,1;"UWATAJ. TWOJE
POLA MAJA NR 7-12": BEEP .2,-10: PAUSE 100: PRINT AT 18,0;"
": GO TO 265

```

```

280>IF w=13THEN LET d=6-b-a(b)
285 IF w=14 THEN LET d=12-b-a(b)
290 GO SUB 500: GO SUB 600: GO TO 250
399 REM * Wypelnienie tabel *
400 FOR i=1 TO 6: LET m(i)=14: LET n(i)=1+4*i: NEXT i
405 FOR i=7 TO 12: LET m(i)=11: LET n(i)=53-4*i: NEXT i
410 LET m(13)=8: LET n(13)=5: LET m(14)=12: LET n(14)=26
420 FOR i=1 TO 12: LET a(i)=4: PRINT AT m(i),n(i);a(i);" ": NEX
T i: RETURN
500 PRINT INK 2; FLASH 1;AT m(b),n(b);a(b)
505 PAUSE 50
510 FOR k=1 TO a(b): LET b=b+1: IF b=c THEN LET b=b+1
515 IF b=13 THEN LET b=1
520 LET a(b)=a(b)+1: PRINT AT m(b),n(b);a(b)
525 NEXT k
530 LET a(c)=0: PRINT AT m(c),n(c);a(c);" "
540 RETURN
599 REM * Zdobyte kamienie *
600 IF d<-6 AND d>-12 THEN LET d=0
610 IF a(b)=2 AND d<=-1 THEN LET pkt=2: GO TO 630
620 IF a(b)=3 AND d<=-1 THEN LET pkt=3: GO TO 630
625 RETURN
630 PRINT INK 2; FLASH 1;AT m(b),n(b);a(b): PAUSE 60: LET a(b)
=0: PRINT AT m(b),n(b);"0 "
635 PRINT INK 2; FLASH 1; AT 6,16;" PREMIA ";(pkt);" PKT. "
640 BEEP .1,7: BEEP .1,11: BEEP .1,14: BEEP .1,17: PAUSE 100: P
RINT AT 6,16;" "
645 IF w=13 THEN LET pktA=pktA+pkt: PRINT AT 15,31-LEN STR$ (p
ktA);pktA
650 IF w=14 THEN LET pktB=pktB+pkt: PRINT AT 10,3-LEN STR$ (pk
tB);pktB
655 IF pktA>=24 OR pktB>=24 THEN GO TO 800
660 LET b=b-1: LET d=d+1
665 IF b=0 THEN LET b=12
670 GO TO 600
799 REM * Koniec kolejnej gry *
800 PRINT INK 2; FLASH 1;AT 6,16;" KONIEC ";(gra);". GRY "
805 IF w=13 THEN PRINT INK 2; FLASH 1;AT 7,16;" WYGRAL: GR-1
"
810 IF w=14 THEN PRINT INK 2; FLASH 1;AT 7,16;" WYGRAL: GR-2
"
820 LET spktA=spktA+pktA: LET spktB=spktB+pktB
830 BEEP .1,12: BEEP .2,16: BEEP .1,7: BEEP .1,14: BEEP .1,16:
BEEP .5,19: PAUSE 300: PRINT AT 6,16;" ";AT 7,16;" "
835 PRINT AT 10,1;" ";AT 15,29;" "
840 PRINT AT 3,30-LEN STR$ spktA;spktA: PRINT AT 4,30-LEN STR$
spktB;spktB
842 INPUT "GRAMY DALEJ (T/N) ";hs
844 IF hs="T" OR hs="t" THEN GO TO 210
846 STOP
849 REM * Polskie litery ACELNOST *
850 POKE USR "A"+7,78: POKE USR "C",8: POKE USR "E"+7,12: POKE
USR "L"+4,112: POKE USR "N",8: POKE USR "O",8: POKE USR "S",8
860 FOR i=0 TO 7: READ n: POKE USR "T"+i,n: NEXT i: RETURN
870 DATA 24,126,4,8,16,32,126,0

```


W każdym numerze >ŁAMIGŁÓWEK< będziemy zamieszczać co najmniej 1 grę zręcznościową >na refleks<.
 Wszystkie gry z instrukcjami po polsku, stosunkowo krótkie, do samodzielnego wpisania do komputera, a zawsze ciekawe i >wciągające<, o różnym stopniu trudności.
 Dziś dla maluchów >CZOŁG NA MOŚCIE<. Pełna instrukcja w listingu programu w liniijkach 6000 - 6060.

Program 9

```

10>GO SUB 9000:GO SUB 6000
15 REM * Inicjacja *
20 INK 0: PAPER 6: BORDER 1: BRIGHT 1: CLS
25 LET x=29: LET cx=0: LET a=1: LET y=160: LET o=10: LET g=15
30 POKE 23296,23279-256*INT (23277/256)
35 POKE 23297,INT (23275/256)
50 LET pkt=0: LET pr=1: LET amu=20: LET utr=1.8: LET s=1
70 POKE 23658,0
80 GO SUB 7000: GO SUB 8000: GO SUB 8020
100 GO TO 1000
500 REM * Ruch czołgu *
505 IF x<1 THEN PRINT AT 5,0;" ": LET pkt=pkt-100
510 IF x<1 THEN GO SUB 8020: GO SUB 5500: LET x=29: RETURN
670 IF s=1 THEN GO TO 2100
675 IF s=0 THEN GO TO 2120
680 RETURN
1000 REM * Główna petla *
1010 IF INT (RND*o)=1 THEN LET pr=0
1100 IF pr=1 THEN GO SUB 500
1110 IF INT (RND*5)=1 THEN LET pr=1
1120 GO SUB 5000
1125 PLOT INVERSE 1:g*8,y+utr
1130 IF y<160 THEN GO SUB 3090
1140 IF INKEY$="p" THEN GO TO 3000
1145 IF INKEY$="z" THEN GO SUB 4000
1150 IF INKEY$="x" THEN GO SUB 4100
1400 GO TO 1000
2090 REM * Druk czołgu *
2100 PRINT AT 5,x;"AB ": LET s=0: RETURN
2120 PRINT AT 5,x-1;"CDE "
2130 LET x=x-1: LET s=1: RETURN
3000 REM * Wystrzał pocisku *
3010 PRINT AT 15,g: INK 2;"G";AT 14,g;"H"
3020 PRINT AT 14,g: INK 2;"H"
3030 RANDOMIZE USR 33000: BORDER 6: PAUSE 2: BORDER 2
3035 LET amu=amu-1: PAUSE 5: BORDER 0: GO SUB 8000
3040 IF amu<0 THEN GO TO 8500
3080 PRINT AT 15,g: PAPER 6;" ":AT 14,g: PAPER 6;" "
3110 PLOT g*8,y: LET y=y-utr: GO TO 1000
4000 IF g<=0 THEN RETURN
4010 LET g=g-1: LET af=(PEEK 23296)-1: POKE 23296,af
4030 RANDOMIZE USR 32000: PRINT AT 16,g;"F": RETURN
4050 POKE 23295,63
4100 IF g>=31 THEN RETURN

```

```

4110>POKE 23296,(PEEK 23296)+1:LET g=g+1
4120 RANDOMIZE USR 32000: PRINT AT 16,g;"F": RETURN
5000 REM * Trafienie ? *
5050 IF y<=128 THEN LET y=160: GO SUB 5240
5060 IF y<=134 THEN GO TO 5100
5070 RETURN
5100 IF g=x OR g=x+1 OR g=x+2 THEN GO TO 5200
5110 RETURN
5210 IF g=x THEN LET pkt=pkt+100
5220 IF g=x+2 THEN LET pkt=pkt+50: LET amu=amu+2
5230 IF g=x+1 THEN LET pkt=pkt+75
5232 GO SUB 8020
5235 PRINT AT 5,0;"
5238 GO SUB 5240: GO TO 5290
5240 PRINT AT 5,g; INK 2;"G";AT 4,g;"H"
5260 BEEP .1,-50
5270 PRINT AT 5,g;" ";AT 4,g;" ": RETURN
5290 LET x=29: LET y=160
5300 IF pkt>=300 THEN LET amu=amu+1
5310 IF pkt>1000 THEN LET o=15: LET utr=utr+1
5320 RANDOMIZE USR 33023: RANDOMIZE USR 33023: RETURN
5500 LET cx=cx+1: PRINT AT 7,0;cx: RETURN
6000 REM * Instrukcja *
6005 BORDER 0:: PAPER 0: INK 7: CLS
6010 REM * STRZELANIE Z ARMATY DO CZOŁGU *
6020 PRINT PAPER 6; INK 9;AT 1,3;"INSTRUKCJA GRY"
6030 PRINT INK 6;AT 5,3;"PRZESUWANIE ARMATY:";AT 11,3;"STRZELANIE DO CZOŁGU:"
6040 PRINT INK 5;AT 7,3; FLASH 1;"W LEWO - klawisz Z";AT 8,3;"W PRAWO - klawisz X";AT 13,3;"OGIEN - klawisz P"
6050 PRINT #0;AT 1,3;"NACISNIJ DOWÓLNY KLAWISZ": PAUSE 0: CLS
6060 RETURN
7000 REM * Grafika ekranu *
7100 PRINT AT 6,0; INK 4;"(6*g8)#####(7*g8)"
7110 PRINT AT 7,0; INK 4;"(6*g8,4*sp,g5,9*sp,g5,4*sp,7*g8)"
7130 PRINT AT 8,0; INK 4;"(7*g9,3*sp,g5,9*sp,g5,3*sp,8*g8)"
7140 PRINT AT 9,0; INK 4;"(8*g8,2*sp,g5)"; INK 5;"(9*g3)"; INK 4;"(g5,2*sp,9*g8)"
7200 PLOT 48,127: DRAW 151,0
7210 PLOT 48,120: DRAW 151,0
7220 PLOT 0,167: DRAW 255,0
7300 PRINT AT 0,0; PAPER 5;"
7400 RANDOMIZE USR 32000: PRINT AT 16,g;"F"
7420 POKE 23295,63: RETURN
7990 REM * Komunikaty *
8000 PRINT AT 0,13; PAPER 5; BRIGHT 1;" AMUNICJA: ";amu;" "
8010 RETURN
8020 PRINT AT 0,2; PAPER 5;"PUNKTY=";pkt;" "
8030 RETURN
8510 POKE 33006,20: POKE 33009,18
8520 PRINT AT 10,11; FLASH 1; INK 2;"KONIEC GRY"
8530 PRINT AT 0,27; FLASH 1; INK 2;"PUSTY"
8535 FOR a=1 TO 3: RANDOMIZE USR 33000: NEXT a
8540 POKE 33019,13: POKE 33004,100: RANDOMIZE USR 33000: POKE 33004,3: POKE 33019,12

```

8545>PRINT AT 10,11;PAPER 5;BRIGHT 1;" UWAGA: ";PAPER 6;"

```
8552 IF pkt>=-1000 THEN LET a$="PRZESPIJ SIE"
8554 IF pkt>=0 THEN LET a$="KIEPSKO"
8556 IF pkt>=100 THEN LET a$="DOSTATECZNIE"
8558 IF pkt>=500 THEN LET a$="DOBRZE"
8560 IF pkt>=1000 THEN LET a$="BARDZO DOBRZE"
8562 IF pkt>=1500 THEN LET a$="IMPONUJACO"
8570 PRINT AT 12,11; INK 1; FLASH 1;a$
8575 POKE 33006,2: POKE 33009,24
8580 PRINT #1;AT 1,4: PAPER 3;"NACISNIJ DOWOLNY KLAWISZ"
8585 IF INKEY$<>"" THEN RUN 2
8590 GO TO 8580
9000 REM * Grafika UDG i progr. masz. *
9010 POKE USR "N",8: POKE USR "S",8: POKE USR "L"+4,112
9020 FOR z=0 TO 38: READ a: POKE 32000+z,a: NEXT z
9030 DATA 33,0,90,6,255,62,63,119,35,16,252,42,0,91,62,0,119,1,3
2,0,237
9040 DATA 66,119,237,66,119,237,66,119,237,66,119,237,66,119,237
,66,119,201
9050 FOR a=0 TO 63: READ z: POKE 65368+a,z: NEXT a
9060 DATA 0,255,7,1,127,255,170,85,112,254,254,248,254,255,170,8
4,0,15,0,0,7,15,5,2,7,255,255
9064 DATA 63,255,255,85,170,0,192,192,0,192,224,64,128,255,165,1
65,165,165,165,165,126,149,215,90
9068 DATA 93,107,50,46,60,129,36,1,138,36,38,88,42
9070 FOR A=0 TO 48: READ X: POKE 33000+A,X: NEXT A
9080 DATA 33,0,0,14,3,22,2,126,230,24,211,254,65,16,254,35,21,32
,244,12,32,239,201,33,0
9084 DATA 64,126,7,119,21,35,124,254,88,32,246,33,0,64,126,15,11
9,35,124,254,88,32,247,201
9090 RETURN
```

F A J K A R Y B A K A

Program dydaktyczny, uczy sumowania liczb ujemnych i dodatnich.
Instrukcja gry w listingu programu.

Program 10

```
10>BORDER 0:PAPER 0:INK 7:CLS
15 CLEAR 63999: RESTORE 9600: FOR i=1 TO 2: GO SUB 9990: NEXT
i
16 RESTORE 8060: GO SUB 8000: GO SUB 820: REM Instrukcja
18 POKE 23658,8: LET spkt=0: LET gra=0
19 REM * Grafika planszy *
20 CLS : PRINT PAPER 6; INK 0;"          F A J K A R Y B A K A
"
25 FOR i=1 TO 5: PRINT PAPER 5;AT i,0;"
": NEXT i
30 PRINT PAPER 5; INK 9;AT 1,0;" W lewo ->L<"" W prawo->P<
"
40 PRINT PAPER 4; INK 0;AT 6,0;"
"
```

```

50>PRINT IN
K 6;AT 16,0;"OBLICZENIA: GRA/PUNKTY: "
60 PRINT AT 16,26; PAPER 6;"(6*g8)"
65 INK 6: PLOT 0,12: DRAW 255,0: PLOT 0,34: DRAW 255,0
70 PRINT PAPER 5;AT 18,0;"
"
80 LET a$="K": LET b$="M": LET c$="Q": LET d$="R"
82 LET e$="N": LET f$="P": LET g$="I": LET h$="U"
90 PRINT PAPER 5; INK 0;AT 5,15;a$: PAPER 4; INK 0;AT 6,15;b$
;AT 7,15;c$
119 REM * Zerowanie gry *
120 LET li=0: LET wl=0: LET wp=0
130 LET gra=gra+1: LET pkt=0: LET suma=0: LET kr2=0
140 PRINT PAPER 5; INK 0;AT 16,27;gra;"/"
499 REM * Inicjacja *
500 PRINT PAPER 2; INK 6;AT 8,0;"(32*g3)"
505 PRINT PAPER 2;AT 9,0;"
"
510 FOR i=0 TO 31: PLOT INK 6;4+i*8,104: DRAW INK 6;0,6: NEXT
i
511 FOR i=9 TO 14: PRINT PAPER 1;AT i,0;"
": NEXT i
520 PLOT INK 7;0,103: DRAW INK 7;255,0
530 FOR i=1 TO 55
535 LET x=INT (RND*5): LET y=INT (RND*31)
540 PRINT PAPER 1; INK 7;AT x+10,y;"B": NEXT i
545 PRINT PAPER 1; INK 7;AT 9,15;"0"
546 BEEP .1,14: BEEP .2,20: INK 7
550 LET zf=INT (RND*2): REM (+ lub -)
552 IF zf=0 THEN LET zf=-1
554 LET mf=INT (RND*120)+80: LET mf=INT (mf/8)*8: LET mf=zf*mf
555 LET x=18: LET y=0: REM Wspolrzedne zapisu rachunku
556 INPUT "PODAJ KIERUNEK (P/L) ? ";k$: IF k$<>"P" AND k$<>"L"
THEN GO TO 556
557 BEEP .05,14: GO SUB 770: GO TO 567
559 REM * Petla gry *
560 IF mf<0 AND suma=INT ((mf-80)/8) THEN GO TO 900
562 IF mf>0 AND suma=INT ((mf+80)/8) THEN GO TO 900
565 INPUT "PODAJ KIERUNEK (P/L) ? ";k$
566 IF k$<>"P" AND k$<>"L" THEN GO TO 565
567 GO SUB 750: INPUT "ILE KROKOW ? (MAX 20) ";r$
568 IF CODE r$<49 OR CODE r$>57 THEN GO TO 567
570 IF LEN r$>2 THEN GO TO 567
571 LET kr=VAL r$: IF kr>20 THEN GO TO 567
572 BEEP .05,20
573 IF k$="P" AND mf>0 AND suma+kr>INT (mf+80)/8+12 THEN GO SU
B 790: GO SUB 785: PAUSE 150: GO SUB 786: GO TO 565
574 IF k$="L" AND mf<0 AND suma-kr<INT (mf-80)/8-12 THEN GO SU
B 790: GO SUB 785: PAUSE 150: GO SUB 786: GO TO 565
575 GO SUB 780: GO SUB 700: REM Wypisywanie obliczen
578 IF wl=1 THEN GO SUB 800
580 IF k$="L" THEN GO TO 600
585 IF k$="P" THEN GO TO 650
599 REM * Przesuwanie wody w prawo *
600 POKE 64002,72: POKE 64004,64
610 FOR i=1 TO kr*8: RANDOMIZE USR 64000
620 LET li=li-1: IF INT (li/8)=li/8 THEN BEEP .01,10: GO SUB 7

```



```

622>LET kr2=kr2+1:IF kr2=4THEN GO SUB 760
624 IF kr2=8 THEN GO SUB 755: LET kr2=0
625 IF k$="L" AND li=mf+40 AND mf<0 THEN PRINT PAPER 1; INK 0
:AT 9,0;"D"
628 IF li=mf+40-250 THEN PRINT PAPER 1; INK 7;AT 9,31;"F"
630 NEXT i
640 IF suma<-40 THEN FOR i=8 TO 14: PRINT INK 6;AT i,0;"(g8)"
: NEXT i: GO SUB 1000: LET w1=1: GO TO 560
645 GO TO 560
649 REM * Przesuwanie wody w lewo *
650 POKE 64025,79: POKE 64027,64
660 FOR i=1 TO 8*kr: RANDOMIZE USR 64023
670 LET li=li+1: IF INT (li/8)=li/8 THEN BEEP .01,10: GO SUB 7
50
672 LET kr2=kr2+1: IF kr2=4 THEN GO SUB 760
674 IF kr2=8 THEN GO SUB 750: LET kr2=0
675 IF k$="P" AND li=mf-48 AND mf>0 THEN PRINT PAPER 1; INK 0
:AT 9,31;"D"
678 IF li=mf-48+248 THEN PRINT PAPER 1; INK 7;AT 9,0;"F"
680 NEXT i
690 IF suma>40 THEN FOR i=8 TO 14: PRINT INK 6;AT i,31;"(g8)"
: NEXT i: GO SUB 1000: LET wp=1: GO TO 560
695 GO TO 560
699 REM * Pisanie rachunku *
700 IF k$="P" THEN PRINT PAPER 5; INK 9;AT x,y:"(+": LET y=y+
2
710 IF k$="L" THEN PRINT PAPER 5; INK 9;AT x,y:"(-": LET y=y+
2
720 PRINT PAPER 5; INK 9;AT x,y;kr:")"+": LET y=y+LEN STR$ kr+2
: IF y>27 THEN LET z=y: LET y=0: LET x=x+1
730 IF k$="P" THEN LET suma=suma+kr
735 IF k$="L" THEN LET suma=suma-kr
745 RETURN
749 REM * Podprogramy *
750 IF k$="P" THEN PRINT PAPER 5; INK 0;AT 5,15;e$; PAPER 4;
INK 0;AT 6,15;f$;AT 7,15;g$
755 IF k$="L" THEN PRINT PAPER 5; INK 0; PAPER 4; INK 0;AT 5,
15;a$;AT 6,15;b$;AT 7,15;c$
758 RETURN
760 IF k$="L" THEN PRINT PAPER 4; INK 0;AT 7,15;d$: REM Noga
765 IF k$="P" THEN PRINT PAPER 4; INK 0;AT 7,15;h$: REM Noga
768 RETURN
780 PRINT AT 9,15;" ": PRINT PAPER 1; INK 7;AT 9,15;"F"
782 RETURN
785 PRINT PAPER 0; INK 5; FLASH 1;AT 21,4;"MUSISZ ZAWROCIC PO
FAJKE": RETURN
786 PRINT AT 21,1;" ": RETURN
790 BEEP .1,14: BEEP .2,20: BORDER 2: PAUSE 5: BORDER 0: RETURN

795 PRINT PAPER 1; INK 7;AT 9,15;"F": RETURN
800 IF wp=1 THEN FOR i=9 TO 14: PRINT PAPER 1;AT i,31;" ": NE
XT i: PRINT PAPER 2; INK 6;AT 8,31;"(g3)": PLOT INK 6;252,104:
DRAW INK 6;0,6: PRINT PAPER 1; INK 7;AT 9,31;"F": LET wp=0
810 IF w1=1 THEN FOR i=9 TO 14: PRINT PAPER 1;AT i,0;" ": NEX
T i: PRINT PAPER 2; INK 6;AT 8,0;"(g3)": PLOT INK 6;4,104: DRA
W INK 6;0,6: PRINT PAPER 1; INK 7;AT 9,0;"F": LET w1=0

```

```

815>RETURN
820 REM * Instrukcja *
825 CLS : PRINT PAPER 6; INK 0;" STARY RYBAK SZUKA SWEJ FAJKI
"
830 PRINT : PRINT " STARY RYBAK JEST BARDZO SMUT-NY, BO ZGUBI
L SWA ULUBIONA FAJKEW BASENIE PORTOWYM."
832 PRINT " NIE WIE W KTORYM KIERUNKU MA JEJ SZUKAC."
834 PRINT " MUSISZ RYBAKOWI DORADZIC W JA-KIM KIERUNKU MA SIE
UDAC I ILE KROKOW MA ZROBIC."
836 PRINT " GDY RYBAK ZNAJDZIE SIE DOKLAD-NIE NAD FAJKA, TO WT
EDY MUSISZ WYLICZYC ILE KROKOW OD PUNKTU 0 ZNAJDOWALA SIE FAJKA
"
840 PRINT " GDY PODASZ PRAWIDLOWY WYNIK TOFAJKA WROCI DO RYBAK
A, A TY O- TRZYMASZ 2 PUNKTY PREMI ZA PIER-WSZA DOBRA ODPOWIEDZ
LUB 1 PUNKTZA DRUGA DOBRA ODPOWIEDZ."
845 PRINT " PO 5 GRACH KOMPUTER OCENI TWO-JE UMIEJETNOSCI LICZ
ENIA."
848 BEEP .1,14: BEEP .2,20
850 PRINT 0; INK 4; FLASH 1;AT 1,4;"Nacisnij dowolny klawisz":
PAUSE 0: BEEP .05,14: RETURN
899 REM * Koniec gry *
910 PRINT AT 21,0;"RYBAK ODNALAZL SWA FAJKE"
920 IF y=0 THEN PRINT PAPER 5; INK 9;AT x-1.z-1;"=";AT x,y;"=
";: LET y=y+1: GO TO 930
925 PRINT PAPER 5; INK 9;AT x,y-1;"=";
930 FOR f=1 TO 3: FOR g=0 TO 7: BEEP .002,g: BORDER g: BEEP .01
,g: NEXT g: NEXT f: BORDER 0
940 INPUT "PODAJ MIEJSCE FAJKI: ";x$
942 IF CODE x$<45 OR CODE x$>57 THEN GO TO 940
944 LET m=VAL x$
946 IF m=suma THEN LET pkt=2: PRINT AT 21,0;"DOBRZE! OTRZYMUJE
SZ 2 PUNKTY": GO SUB 795: GO TO 960
950 INPUT "ZLE! SPROBUJ JESZCZE RAZ: ";x$
952 IF CODE x$<45 OR CODE x$>57 THEN GO TO 950
954 LET m=VAL x$
956 IF m=suma THEN LET pkt=1: PRINT AT 21,0;"DOBRZE! OTRZYMUJE
SZ 1 PUNKT": GO SUB 795: GO TO 960
958 PRINT #0;AT 1,0;"ZLE! WYNIK WYNOSI ":suma
960 GO SUB 790: GO SUB 795: PRINT PAPER 6; INK 9; FLASH 1;AT x
,y;suma: LET spkt=spkt+pkt
962 PRINT PAPER 5; INK 0;AT 16,27;gra:"/";spkt: PAUSE 5
965 IF gra=5 THEN PAUSE 0: GO TO 980
970 INPUT "GRAMY JESZCZE RAZ ? ";v$
972 IF v$="T" AND gra=5 THEN GO TO 18
974 IF v$="T" THEN GO TO 20
976 IF v$="N" THEN STOP
978 GO TO 970
980 IF spkt>=7 THEN LET w$="DOBRY ! "
982 IF spkt>=9 THEN LET w$="BARDZO DOBRY !"
984 IF spkt<7 THEN LET w$="DOSTATECZNY ! "
986 GO SUB 1050: PRINT AT 21,0;"WYNIK W 5 GRACH: "; INK 3; BRIG
HT 1;;w$: REM Ocena 5 gier
988 PRINT PAPER 5; INK 0;AT 16,27;gra:"/";spkt
990 FOR f=1 TO 5: FOR g=0 TO 7: BEEP .02,g: BORDER g: NEXT g: N
EXT f: BORDER 0
995 GO TO 970

```

```

999>REM * Komunikaty *
1000 PRINT #0; FLASH 1; AT 1,0;" KONIEC BASENU - MUSISZ WRACAC "
1005 PAUSE 0: GO SUB 1050: RETURN
1050 PRINT #0; AT 1,0;"          ": RETURN
7999 REM * Tworzenie grafiki *
8000 FOR i=0 TO 7: READ z: POKE USR "B"+i,z: NEXT i
8010 FOR i=0 TO 7: READ z: POKE USR "D"+i,z: NEXT i
8020 FOR i=0 TO 7: READ z: POKE USR "F"+i,z: NEXT i
8030 FOR i=0 TO 7: READ z: POKE USR "K"+i,z: NEXT i
8035 FOR i=USR "M" TO USR "N"+7: READ z: POKE i,z: NEXT i
8040 FOR i=0 TO 7: READ z: POKE USR "P"+i,z: NEXT i
8045 FOR i=USR "Q" TO USR "R"+7: READ z: POKE i,z: NEXT i
8048 FOR i=0 TO 7: READ z: POKE USR "I"+i,z: NEXT i
8050 FOR i=0 TO 7: READ z: POKE USR "U"+i,z: NEXT i: RETURN
8070 DATA 0,0,96,145,14,0,0,0
8080 DATA 255,15,12,236,236,236,252,124,255,0,0,0,0,0,0
8110 DATA 30,62,255,46,98,50,20,60,30,26,58,106,74,75,107,59
8120 DATA 120,124,255,116,70,76,40,60,120,88,92,86,82,210,214,22
0
8130 DATA 30,28,28,28,12,12,12,28,30,62,54,119,103,99,99,231
8140 DATA 120,56,56,56,48,48,48,56,120,124,108,238,230,198,198,2
31
9601 DATA 64000,9601
9602 DATA "2100400EC00620E5DD17"
9603 DATA "E1DD7E1F1FCB1E231096"
9604 DATA "FBOD20EFC9E0",""
9611 DATA 64023,9611
9612 DATA "21FF570EC00620E5DD2D"
9613 DATA "E1DD7EE117CB162B1050"
9614 DATA "FBOD20EFC9E0",""

```

c.d. str.16

ROZWIĄZANIE ZADAŃ

ZADANIE DLA KOMPUTERA (RAZ * 12 = TUZIN)	897*12=10762
LICZBY FIBONACCIEGO - liczba 39-ta	63245985
KOMPUTRUS - dla 32 matematyków i 7 kroków	5 miejsce
TEORIA WZGLĘDNOŚCI I MASA	299000 km/sek
FLAGA PAŃSTWA AFRYKAŃSKIEGO	TOGO
MELODIA JEDNOGŁOSOWA	Krakowiaczek ci ja
ZADANIE TEKSTOWE DLA KOMPUTERA	114, 115, 116, 200, 212

W NASTĘPNYM NUMERZE

ŁAMIGŁÓWKI MATEMATYCZNE - ZGADYWANKI - 6 zadań
 ARIADNA - Gra planszowa
 OBRONA PRZED RAKIETAMI - Gra zręcznościowa
 SZYBKIE DODAWANIE - Dydaktyka szkolna
 PISZEMY PROGRAM DLA SIEBIE - Znaki Zodiaku
 CIEKAWA GRAFIKA - Krakowiana

ADRES REDAKCJI: 30-106 Kraków, ul. Senatorska 25/11
 REDAKTOR NACZELNY: Tadeusz Binek tel. (0-12) 22-36-39
 WYDAWCA: Zakład Usług Informatycznych >BETA< w Krakowie
 DRUK: ZWR >AGROBISP<, Kraków, Plac Na Stawach 3

```

9700>READ a,n
9710 READ l$: LET l=LEN l$: LET s=0: LET k=2: LET n=n+1
9720 IF l=0 THEN RETURN
9730 LET q2=CODE l$(K-1): LET q1=CODE l$(k)
9740 LET c=q1-48-7*(q1>64)+16*(q2-48-7*(q2>64))
9750 IF k<l THEN POKE a,c: LET s=s+c: LET k=k+2: LET a=a+1: GO
TO 9730
9760 GO TO 9710

```

C I E K A W A G R A F I K A

Program 11

```

500>BORDER 1:PAPER 1:INK 6:CLS
502 LET a=148: LET b=149: LET c=150: LET d=151: LET e=152: LET
f=125: LET g=118
510 PLOT 120,e: DRAW 2,0: PLOT 118,d: DRAW 7,0: DRAW 0,-1: DRAW
-7,0: DRAW 0,-1: DRAW 7,0: DRAW 0,-1: DRAW -7,0
512 LET i=0
514 PLOT f,c: DRAW 16,20: DRAW 27,-12
515 PLOT g,c: DRAW -16,20: DRAW -27,-12
516 PLOT f,b: DRAW 23,15: DRAW 23,-16
517 PLOT g,b: DRAW -23,15: DRAW -23,-16
518 PLOT f,a: DRAW 26,8: DRAW 18,-27
519 PLOT g,a: DRAW -26,8: DRAW -18,-27
520 LET i=i+2
522 READ n: BEEP .1,n: IF i/56=INT (i/56) THEN RESTORE 550
523 INVERSE 1: PLOT 120,e+2-i: DRAW 2,0: PLOT 118,e+1-i: DRAW 7
,0: PLOT 118,e-i: DRAW 7,0: INVERSE 0: PLOT 120,e-i: DRAW 2,0: P
LOT 118,b-i: DRAW 7,0: DRAW 0,-1: DRAW -7,0
524 INVERSE 1: PLOT f,e-i: DRAW 16,20: DRAW 27,-12
525 PLOT g,e-i: DRAW -16,20: DRAW -27,-12
526 INVERSE 0: PLOT f,b-i: DRAW 28,0: DRAW 0,-30
527 PLOT g,b-i: DRAW -28,0: DRAW 0,-30
528 INVERSE 1: PLOT f,b-i: DRAW 28,0: DRAW 0,-30
529 PLOT g,b-i: DRAW -28,0: DRAW 0,-30
530 INVERSE 0: PLOT f,c-i: DRAW 16,20: DRAW 27,-12
531 PLOT g,c-i: DRAW -16,20: DRAW -27,-12
534 INVERSE 1: PLOT f,d-i: DRAW 23,15: DRAW 23,-16
535 PLOT g,d-i: DRAW -23,15: DRAW -23,-16
536 INVERSE 0: PLOT f,b-i: DRAW 23,15: DRAW 23,-16
537 PLOT g,b-i: DRAW -23,15: DRAW -23,-16
540 INVERSE 1: PLOT f,c-i: DRAW 26,8: DRAW 18,-27
541 PLOT g,c-i: DRAW -26,8: DRAW -18,-27
542 INVERSE 0: PLOT f,a-i: DRAW 26,8: DRAW 18,-27
543 PLOT g,a-i: DRAW -26,8: DRAW -18,-27
545 IF i=112 THEN GO TO 200
548 GO TO 520
550 DATA 0,5,9,7,5,4,5,0,5,9,7,5,4,5
552 DATA 0,5,9,12,10,9,7,0,4,7,10,9,7,5

```

